/****************************************************************************
 **
 ** Copyright (C) 2011 Nokia Corporation and/or its subsidiary(-ies).
 ** All rights reserved.
 ** Contact: Nokia Corporation (qt-info@nokia.com)
 **
 ** This file is part of the examples of the Qt Toolkit.
 **
 ** $QT_BEGIN_LICENSE:BSD$
 ** You may use this file under the terms of the BSD license as follows:
 **
 ** "Redistribution and use in source and binary forms, with or without
 ** modification, are permitted provided that the following conditions are
 ** met:
 **   * Redistributions of source code must retain the above copyright
 **     notice, this list of conditions and the following disclaimer.
 **   * Redistributions in binary form must reproduce the above copyright
 **     notice, this list of conditions and the following disclaimer in
 **     the documentation and/or other materials provided with the
 **     distribution.
 **   * Neither the name of Nokia Corporation and its Subsidiary(-ies) nor
 **     the names of its contributors may be used to endorse or promote
 **     products derived from this software without specific prior written
 **     permission.
 **
 ** THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS
 ** "AS IS" AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT
 ** LIMITED TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR
 ** A PARTICULAR PURPOSE ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT
 ** OWNER OR CONTRIBUTORS BE LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL,
 ** SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT
 ** LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES; LOSS OF USE,
 ** DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND ON ANY
 ** THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT
 ** (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE
 ** OF THIS SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE."
 ** $QT_END_LICENSE$
 **
 ****************************************************************************/

 import QtQuick 1.0

 Flipable {
     id: container

     property alias image: frontImage.source
     property bool flipped: true
     property int xAxis: 0
     property int yAxis: 0
     property int zAxis: 0
     property int angle: 0

     width: front.width; height: front.height

     front: Image { id: frontImage; smooth: true }
     back: Image { source: "back.jpg"; smooth: true }

     state: "back"

     MouseArea { anchors.fill: parent; onClicked: container.flipped = !container.flipped }

     transform: Rotation {
         id: rotation; origin.x: container.width / 2; origin.y: container.height / 2
         axis.x: container.xAxis; axis.y: container.yAxis; axis.z: container.zAxis
     }

     states: State {
         name: "back"; when: container.flipped
         PropertyChanges { target: rotation; angle: container.angle }
     }

     transitions: Transition {
         ParallelAnimation {
             NumberAnimation { target: rotation; properties: "angle"; duration: 1000 }
             SequentialAnimation {
                 NumberAnimation { target: container; property: "scale"; to: 1.0; duration: 100 }
                 NumberAnimation { target: container; property: "scale"; to: 1.0; duration: 300 }
             }
         }
     }
 }
